ナルシシスト数について

岡本　茂

On the Narcissistic Number

Shigeru OKAMOTO

abstract

This paper is devoted to generalization of narcissistic number. Let n be a natural number and $n = \Sigma 10^i q_i$ be the decimal representation, and $p>1$ be a natural number. Then, n is called a generalized narcissistic number for p if $\Sigma q_i^p = n$, and p is called as index of narcissistic number.

We prove the following theorems:

Theorem 1. Narcissistic numbers with index 3 are 1,153,370,371 and 407.

Theorem 2. Number of generalized narcissistic numbers with index p are finite.

In the proof of Theorem 2, we have the following inequality:

$$(n\text{-}log\,(n+1))/log\ 9 < n$$

,which concerns an upper bound of index p.

1．はじめに

これは，参考文献［1］にある問題の完全解をまとめ，それを一般化したものである．［1］におけるナルシシスト数の定義は単純で，すでに完全に求められている．ここでは[1]にある解答を補い，それを明確にし，一般化することを目標とした．

プログラムは BASIC によって作ったが，プログラム構造についてはある程度配慮したので，これをC言語などに翻訳することはたいへん容易である．ここでは長桁計算を使い，具体的に $0^p, 1^p, \dots, 9^p$ を $p \leqq 100$ で計算したが，この計算では伊藤 美和さんにたいへんお世話になった．ここに厚く感謝する．

2．ナルシシスト数について

正の整数で，その各桁の数を3乗して加えた結果がもとの数にもどるものを，ナルシシスト数（Narcissistic Cube)という．それを求めるプログラムは，たとえば次のようになる．ここでは10000以下の正整数を対象とした．このプログラムをプログラムとして引用する．

```
100 'The Narcissistic Cube
```

```
110 DEFINT I-Z
120 DIM P(100), X(5)
130 READ K,L               '求めている範囲の入力
140 J=0
150 FOR N=K TO L
160     M=N
170         FOR I=1 TO 5:X(I)=M MOD 10:M=M ¥ 10:NEXT
180         K=0:FOR I=1 TO 5:L=X(I):K=K+L*L*L:NEXT
190     IF N=K THEN J=J+1:P(J)=N
200 NEXT
210 PRINT "ナルシシスト数は次の通りです：";:FOR K=1 TO J:PRINT P(K);:NEXT
220 END
300 DATA 1,10000
```

これで実験すると，ナルシシスト数は1,153,370,371,407の5個になる．最初に，ナルシシスト数はこれ以外にないことを証明する．

定理Ⅰ　ナルシシスト数は1,153,370,371,407だけである．

証明　1,153,370,371,407がナルシシスト数になることは，直接計算で明らかである．これ以外にないことを示そう．最初に

$$m=10^{n}q_n+10^{n-1}q_{n-1}+\cdots+10q_1+q_0=\Sigma 10^{i}q_i \qquad (1)$$

を$n+1$桁の任意の正整数とし，$n>3$ならばmはナルシシスト数ではないことを示す．ここで

$$0\leqq q_i\leqq 9,\ (i=0,1,\cdots,n) \qquad (2)$$

で，q_iは第n-i桁の数を表わす．したがって，$q_n\geqq 1$と仮定してよく，(1)より直ちに $10^{n+1}>m\geqq 10^{n}$ が得られる．また，(2)を次のように書き改めることができる．

$$0\leqq q_i\leqq 9,\ (i=0,1,\cdots,n-1),\ 1\leqq q_n\leqq 9 \qquad (3)$$

まずmのナルシシスト値$v(m)$を次のように定義する．

$$v(m)=q_n^3+q_{n-1}{}^3+\cdots+q_1^3+q_0^3$$

このとき，ナルシシスト数は $m=v(m)$ となる数である．

$$\begin{aligned}m-v(m)&=(10^{n}q_n+10^{n-1}q_{n-1}+\cdots+10q_1+q_0)-(q_n^3+q_{n-1}{}^3+\cdots+q_1^3+q_0^3)\\&=(10^{n}q_n-q_n{}^3)+(10^{n-1}q_{n-1}-q_{n-1}{}^3)+\cdots\\&\qquad+(10q_2-q_2^3)+(10q_1-q_1^3)+(q_0-q_0^3)\\&=(10^{n}-q_n{}^2)q_n+(10^{n-1}-q_{n-1}{}^2)q_{n-1}+\cdots\\&\qquad+(10^2-q_2^2)q_2+(10-q_1^2)q_1+(1-q_0^2)q_0\end{aligned}$$

ここで(2)により，

$$n\text{-}1\geqq i\geqq 2 \quad ならば \quad (10^{i}-q_i{}^2)q_i\geqq 0$$

となるから

$$m-v(m) \geqq 10^{n}-q_n^2+(10^2-q_2^2)q_2+(10-q_1^2)q_1+(1-q_0^2)q_0$$

また，q_1, q_0 とも 9 を超えないから

$$10-q_1^2 \geqq -71,\quad 1-q_0^2 \geqq -80$$

となる．ここで q_n も 9 を超えないから

$$\begin{aligned} m-v(m) &\geqq 10^{n}-81-71\times 9-80\times 9 \\ &=10^{n}-1440 \\ &\geqq 10000-1440 \\ &=8560>0 \end{aligned}$$

よって，$n>3$ ならば m はナルシシスト数ではない．$n=0,1,2,3$では直接計算するのが最も簡単で，それは上のプログラムを実行すればよい．　　　　　　（証明終）

上記証明をもう少し精密にやれば，$n>3$という条件を狭くできるが，それはここでは取り上げない．

3．ナルシシスト数の一般化

ナルシシスト数では指数を3とした．ここでは指数を一般化することにより，ナルシシスト数を一般化する．

定義 3.1　一般ナルシシスト値

正整数 $m=10^{n}q_n+10^{n-1}q_{n-1}+\cdots+10q_1+q_0=\Sigma 10^{i}q_i$ の正整数 $p>1$ に対する一般ナルシシスト値を

$$v(m,p)=q_n{}^p+q_{n-1}{}^p+\cdots+q_1{}^p+q_0{}^p$$

で定める．p を一般化ナルシシスト値の指数という

定義3.2　一般ナルシシスト数

一般ナルシシスト値がもとの正整数に等しいとき，この正整数を一般ナルシシスト数という．

これについても，次のように上記定理Ⅰが一般化できる．

定理Ⅱ　指数 p に対する一般ナルシシスト数は有限個である．

証明　ある正整数 k が存在して，$\Sigma 10^{i}q_i > \Sigma q_i{}^p$ *for* $n>k$ となることを示せばよく，

$$\Sigma 10^{n} > 9^{p}(n+1) \qquad (4)$$

となる最小の n を k と置けばよい． この不等式の左辺は n の指数関数で，右辺は n の 1次関数だから，このような k は確かに存在する．このとき，$q_n \geqq 1$ に注意すれば

$$\begin{aligned} \Sigma 10^{i}q_i &\geqq 10^{n}q_n \\ &\geqq 10^{n} \\ &> 9^{n}(n+1) \geqq \Sigma q_i{}^p \quad (\because\ 9^{p} \geqq q_i{}^p \ \ for\ n \geqq i \geqq 0) \end{aligned}$$

（証明終）

具体的に一般ナルシシスト数を求めることは，それほど難しくないが，指数が大きくなると大変である．ここでは，指数が8までの場合について具体的に求めたので，それを示す．

指数	解の個数	具 体 的 な 解
3	5	1,153,370,371,407
4	4	1,1634,8208,9434
5	7	1,4150,4151,54748,92727,93084,194979
6	2	1,548834
7	6	1,1741725,4210818,9800817,9926315,14459929
8	4	1,24678050,24678051,88593477

$p=2$ では一般ナルシシスト数は 1 だけで，これは直接証明することができる．また，プログラムは前節に引用したものを大幅に修正し，整数型変数だけを使い，いわゆる長桁計算により求めた．プログラムを一般化するためには，ある程度の工夫が必要である．すなわち，pを大きくしたときは9^pなどが相当大きくなるから，長桁計算を考える必要がある．ここでは整変数が表わす範囲を 0 から9999までに限定し，10000以上の整数は整変数による配列で表わした．たとえば，次のプログラムでは指数を5として計算したが，ここでは，$0^5, 1^5, \ldots, 9^5$ を具体的に配列として計算しておき，$I^5 = P(I,1) \times 10000 + P(I,2)$ となっている．

また，それに基づいて一般ナルシシスト値 $v(0,p)$ から $v(9999,p)$ を求め，それをファイルにまとめておくと便利である．ファイルを使うと，一般ナルシシスト数でかなり有効で，計算がかなり自動的にできる．そのプログラムも参考までに示した．

プログラム②

```
100 'The Narcissistic number with index 5
110 DEFINT A-Z
120 'There exist 6 solutions 4150,4151,54748,92727,93084 and 194979.
130 DIM P(9,2)
140 FOR I=0 TO 9:READ P(I,1),P(I,2):NEXT
150 READ A,B
160 FOR X1=A TO B
170     PRINT USING "####:";X1;
180     Z=X1:Y1=0:Y2=0
190     WHILE Z>0
200         U=Z MOD 10:Y1=Y1+P(U,1):Y2=Y2+P(U,2):Z=Z ¥ 10
210         IF Y2>9999 THEN Y1=Y1+1:Y2=Y2-10000
220     WEND
230     FOR X2=0 TO 9999
```

```
240        Z=X2:Z1=Y1:Z2=Y2
250        WHILE Z>0
260            U=Z MOD 10:Z1=Z1+P(U,1):Z2=Z2+P(U,2):Z=Z ¥ 10
270            IF Z2>9999 THEN Z1=Z1+1:Z2=Z2-10000
280        WEND
290        IF Z2=X2 AND Z1=X1 THEN PRINT USING "#### ";X2;
300    NEXT X2
310    PRINT
320 NEXT X1
330 END
500 '5乗のデータ
510 DATA 0,0         '=0↑5
520 DATA 0,1         '=1↑5
530 DATA 0,32        '=2↑5
540 DATA 0,243       '=3↑5
550 DATA 0,1024      '=4↑5
560 DATA 0,3125      '=5↑5
570 DATA 0,7776      '=6↑5
580 DATA 1,6807      '=7↑5
590 DATA 3,2768      '=8↑5
600 DATA 5,9049      '=9↑5
700 DATA 0,100
```

指数が4以下なら，配列Pは1次元でよく，このプログラムが少し簡単になる．次に，ファイルを使ったプログラムを示す．P7__2.TBL は一般ナルシシスト値 $v(0,7)$ から $v(9999,7)$ までを記録したファイルで，0^7から9^7までが計算できていれば容易に作成できる．

```
100 'The Narcissistic number with index 7
110 DEFINT A-Z
120 'There exist 6 solutions 4150,4151,54748,92727,93084 and 194979.
130 DIM P(9999,2)
200 OPEN "F:P7__2.TBL" FOR INPUT AS #1
210       FOR I=0 TO 9999:INPUT P(I,1),P(I,2):NEXT
220 CLOSE #1
300 FOR X1=0 TO 9999
310     PRINT USING "####:";X1;
320     Y1=P(X1,1):Y2=P(X1,2)
330     IF Y1>X1 THEN 500
400     FOR X2=0 TO 9999
410         Z1=Y1+P(X2,1):Z2=Y2+P(X2,2)
420         Z1=Z1 ¥ 10000:Z2=Z2 MOD 10000
```

```
430         IF Z2=X2 AND Z1=X1 THEN PRINT USING "#### ####";X1,X2
440       NEXT X2
500 NEXT X1
600 END
```

本来は行120の結果をファイルに収容すべきであろうが,それは別のプログラムで行なった. また，ここでは，0から99999999までの8桁以下の正整数を対象として計算したが，次説でその十分なことを示すことができる.

4．不等式 $10^n > 9^p(n+1)$ について

定理IIの証明における不等式(4)は，Gaussの記法を使って

$$p \leqq [(n - \log(n+1))/\log 9] \qquad (5)$$

と書き直すことができる．ここで $\log$ は常用対数を表わし，右辺の [ ] は整数部分を表わし，その内部は n の単調増加関数である．これを次のように表わすとわかりやすい.

定理III 不等式(5)をみたす n と p について $p \leqq n-1$ が成り立つ.

証明 $f(n) = (n - \log(n+1))/\log 9$ で，n を 1 以上の実数と考えて変化を調べればよい. $f'(n) = 1/\log 9 - \log e/((n+1)\log 9)$ だから，$n \geqq 1$で $1 > f'(n) > 0$がわかる．これから，$n > 1$で $n > f(n)$ が確かめられる. (証明終)

ここで，具体的に(5)の右辺を数値計算すれば，上記補助定理における不等式の意味が多少明らかになる．これは一種の安定性を示すものである．その結果の一部を示す.

$n \geqq 1$ ならば $p \leqq n-1$, $n \geqq 34$ ならば $p \leqq n$, $n \geqq 60$ ならば $p \leqq n+1$,
$n \geqq 84$ ならば $p \leqq n+2$, $n \geqq 108$ ならば $p \leqq n+3$, $n \geqq 130$ ならば $p \leqq n+4$,
$n \geqq 153$ ならば $p \leqq n+5$, $n \geqq 175$ ならば $p \leqq n+6$, $n \geqq 197$ ならば $p \leqq n+7$,
$n \geqq 218$ ならば $p \leqq n+8$, $n \geqq 240$ ならば $p \leqq n+9$, $n \geqq 262$ ならば $p \leqq n+10$.

これは，n を大きくしていくと p の許容範囲が少しずつひろがっていくことを示しており，数値計算を行なう場合には意味があるかも知れない.

参考文献

[1] Stephen L.Snover & Mark A.Spikell : Brain Ticklers, 1981, Prentice-Hall Inc., 玄 光男，平井 伝逸訳 パズルとマイコン, 1983, 共立出版㈱